combinatorial differences of chains

Johan G. F. Belinfante
2011 April 15

```
In[1]:= SetDirectory["l:"]; << goedel.11apr14a

:Package Title: goedel.11apr14a                2011 April 14 at 5:05 p.m.

It is now:  2011 Apr 15 at 16:31

Loading Simplification Rules

TOOLS.M is now incorporated in the GOEDEL program as of 2010 September 3

weightlimit = 40
```

summary

If **x** and **y** are chains of sets, then **image[DIF, x × y]** is closed under binary intersections, and therefore is a basis for a topology on **U[x] – A[x]**.

derivation

Theorem. The class **image[DIF, spine[S, x] × spine[S, y]]** is closed under binary intersections.

```
In[2]:= ImageComp[CAP, composite[cross[DIF, DIF], TWIST],
          cart[cartsq[spine[S, x]], cartsq[spine[S, y]]]] // Reverse

Out[2]= image[CAP, cart[image[DIF, cart[spine[S, x], spine[S, y]]],
            image[DIF, cart[spine[S, x], spine[S, y]]]]] ==
         image[DIF, cart[spine[S, x], spine[S, y]]]

In[3]:= image[CAP, cart[image[DIF, cart[spine[S, x_], spine[S, y_]]],
            image[DIF, cart[spine[S, x_], spine[S, y_]]]]] :=
         image[DIF, cart[spine[S, x], spine[S, y]]]
```

Corollary. (Eliminating the **spine** wrapper.)

```
In[5]:= SubstTest[implies, and[equal[x, spine[S, u]], equal[y, spine[S, v]]],
          equal[image[CAP, cartsq[image[DIF, cart[x, y]]]], image[DIF, cart[x, y]]],
          {u → x, v → y}] // Reverse

Out[5]= or[equal[image[CAP, cart[image[DIF, cart[x, y]], image[DIF, cart[x, y]]]],
            image[DIF, cart[x, y]]], not[subclass[cart[x, x], union[S, inverse[S]]]],
          not[subclass[cart[y, y], union[S, inverse[S]]]]] == True
```

```
In[7]:= or[equal[image[CAP, cart[image[DIF, cart[x_, y_]], image[DIF, cart[x_, y_]]]],
           image[DIF, cart[x_, y_]]], not[subclass[cart[x_, x_], union[S, inverse[S]]]],
         not[subclass[cart[y_, y_], union[S, inverse[S]]]]] := True
```

eliminating variables

Theorem.

```
In[8]:= SubstTest[member, U[t], V, t → image[DIF, cart[setpart[x], y]]]

Out[8]= member[image[DIF, cart[setpart[x], y]], V] == True

In[9]:= member[image[DIF, cart[setpart[x_], y_]], V] := True
```

Corollary.

```
In[10]:= SubstTest[member, image[DIF, cart[setpart[t], y]],
           V, t → spine[S, setpart[x]]] // Reverse

Out[10]= member[image[DIF, cart[spine[S, setpart[x]], y]], V] == True

In[11]:= member[image[DIF, cart[spine[S, setpart[x_]], y_]], V] := True
```

Lemma. (Remove the **spine** wrappers.)

```
In[12]:= SubstTest[implies, and[equal[x, spine[S, setpart[u]]], equal[y, spine[S, v]]],
           member[image[DIF, cart[x, y]], binclosed[CAP]], {u → x, v → y}] // Reverse

Out[12]= or[and[member[image[DIF, cart[x, y]], V], subclass[image[CAP,
              cart[image[DIF, cart[x, y]], image[DIF, cart[x, y]]]], image[DIF, cart[x, y]]]],
           not[member[x, V]], not[subclass[cart[x, x], union[S, inverse[S]]]],
           not[subclass[cart[y, y], union[S, inverse[S]]]]] == True

In[13]:= (% /. {x → x_, y → y_}) /. Equal → SetDelayed
```

Theorem. (Eliminate the variables **x** and **y**.)

```
In[14]:= Map[empty[composite[Id, complement[#]]] &, SubstTest[class, pair[x, y],
           implies[and[member[x, u], member[y, u]], member[pair[x, y], v]],
           {u → chains[S], v → image[inverse[composite[IMAGE[DIF], CART]], binclosed[CAP]]}]]

Out[14]= subclass[image[IMAGE[DIF], image[CART, cart[chains[S], chains[S]]]], binclosed[CAP]] ==
          True

In[15]:= subclass[
           image[IMAGE[DIF], image[CART, cart[chains[S], chains[S]]]], binclosed[CAP]] := True
```

Corollary. Combinatorial differences of chains are bases for topologies.

```
In[16]:= SubstTest[implies, subclass[u, v], subclass[image[t, u], image[t, v]],
           {t → UCLOSURE, u -> image[IMAGE[DIF], image[CART, cartsq[chains[S]]]],
            v -> binclosed[CAP]}] // Reverse

Out[16]= subclass[image[UCLOSURE,
            image[IMAGE[DIF], image[CART, cart[chains[S], chains[S]]]]], TOPS] == True

In[17]:= subclass[image[UCLOSURE,
            image[IMAGE[DIF], image[CART, cart[chains[S], chains[S]]]]], TOPS] := True
```

reintroducing variables

To obtain a topology, it suffices for **x** to be a set, whereas **y** can be any class.

Lemma.

```
In[18]:= SubstTest[implies, member[t, binclosed[CAP]], member[Uclosure[t], TOPS],
           t → image[DIF, cart[spine[S, setpart[x]], spine[S, y]]]] // Reverse

Out[18]= member[Uclosure[image[DIF, cart[spine[S, setpart[x]], spine[S, y]]]], TOPS] == True

In[19]:= member[Uclosure[image[DIF, cart[spine[S, setpart[x_]], spine[S, y_]]]], TOPS] := True
```

Theorem. If **x ∈ chains[S]** and **P[y] ⊂ chains[S]**, then **Uclosure[image[DIF, x × y]] ∈ TOPS**.

```
In[20]:= Map[implies[member[x, z], #] &,
           SubstTest[implies, and[equal[x, spine[S, setpart[u]]], equal[y, spine[S, v]]],
            member[Uclosure[image[DIF, cart[x, y]]], TOPS], {u → x, v → y}]] // Reverse

Out[20]= or[member[Uclosure[image[DIF, cart[x, y]]], TOPS],
           not[member[x, z]], not[subclass[cart[x, x], union[S, inverse[S]]]],
           not[subclass[cart[y, y], union[S, inverse[S]]]]] == True

In[21]:= or[member[Uclosure[image[DIF, cart[x_, y_]]], TOPS],
           not[member[x_, z_]], not[subclass[cart[x_, x_], union[S, inverse[S]]]],
           not[subclass[cart[y_, y_], union[S, inverse[S]]]]] := True
```

Comment. This topology is a topology on the set **U[x] – A[y]**.

```
In[22]:= U[Uclosure[image[DIF, cart[x, y]]]]

Out[22]= intersection[complement[A[y]], U[x]]
```